

Paper 224-2008

PROC REPORT in *Color ... What's Your STYLE?*

Wendy Boberg, Arkansas Foundation for Medical Care, Little Rock, Arkansas

ABSTRACT

Are you using the Output Delivery System (ODS)? Have you found the STYLE, layout and color scheme, you like? Do you know how to edit the colors on a table using ODS PDF? Do you know how to add background color to highlight a row, column, or cell? I used to get a highlighter out of my desk and go to work, but now I let SAS® and the printer do all of the work.

This paper will discuss a few different ways you can customize your reports. I will share my favorite styles and show you how to look at your tables in the different styles available. I will briefly explain how you can create your own style by changing the color scheme of an existing style so that it can be used for all of your reports. You will learn by example how to truly customize your table using PROC REPORT. I will explain how I used the pieces of code to get the table with a row, a column and individual cells highlighted with several different colors.

INTRODUCTION

This paper provides examples for creating PDF files using ODS, but these methods can be used with other output destinations. I will share what styles I like to use, and how you can edit styles to change the appearance of your reports. In addition, I will discuss a couple of different methods to customize reports by adding colors with the STYLE option. This paper is for beginner to intermediate level programmers with experience creating tables with the REPORT procedure.

THE STYLE OPTION

If you use the ODS to create a PDF file, make sure you specify a STYLE. If you do not know what styles will work for you, ask yourself these questions: 1) Is there a particular color scheme that should be used or avoided? 2) Are you only creating a stand-alone table, or are you creating a report that consists of a table and graphs? 3) If you also have a graph, should the table have the same color scheme as the graph? 4) How will the report be used? 5) Will the report be viewed electronically? 6) Will the report be printed in black and white? 7) Will the report be inserted into a presentation?

I recommend that you take some time to look at different styles to determine your favorite ones. There are several styles available. To get a list, run this TEMPLATE procedure.

```
PROC TEMPLATE;
  LIST styles;
RUN
```

If you have never created an ODS PDF file, try running the code below, and replace the underline with the name of the style you want to see (for the default style, either do not include the style= option or specify style=default):

```
ODS LISTING CLOSE;
ODS PDF NOTOC STYLE= _____
      FILE="C:\TEMP\TESTING _____ STYLE.PDF";
TITLE1 J=C "This is the _____ STYLE";
FOOTNOTE1 J=L "This is just a test";
      Insert PROC REPORT (or PROC PRINT OR PROC SQL) statement.
ODS _ALL_ CLOSE;
ODS LISTING;
```

If you plan to create a report that includes graphics, I recommend reading the SUGI30 paper by Jeff Carter, titled [Use of Styles in Graphics](#). In his paper, he lists the 16 ODS-supplied styles that are ready to be applied to SAS/GRAPH®, SAS/STAT®, and SAS/ETS®. When I create a report with graphics, I typically use the SASWEB style for the table and the STATISTICAL or ANALYSIS style for the plot; however, there are times that I will use the STATISTICAL or ANALYSIS style for both the table and the plot. These are just my

preferences. There are several styles to choose from. Figures 1-3 show some examples of these STYLES applied to a small table I created with the PRINT procedure.

FIGURE 1: SASWEB

This is the SASWEB STYLE

Obs	NAME	TITLE_DEPT
1	Wendy	Statistician
2	Amy	Statistician

FIGURE 2: STATISTICAL

This is the STATISTICAL STYLE

Obs	NAME	TITLE_DEPT
1	Wendy	Statistician
2	Amy	Statistician

FIGURE 3: ANALYSIS

This is the ANALYSIS STYLE

Obs	NAME	TITLE_DEPT
1	Wendy	Statistician
2	Amy	Statistician

PROC TEMPLATE

Did you find a perfect style? If you find a style that is almost perfect but want to change a few things, like the background color, consistently in all your reports, then I recommend using the TEMPLATE procedure. If the color blue is the only thing you do not like about the SASWEB style, then you can edit and create your own style with PROC TEMPLATE. To create your own style, essentially all you do is get a copy of the style's code and paste it into the editor window and then create a new template using this code.

To get a copy of the style's code, right click on Results and select Templates. Then expand the SASHELP.TMPLMST node and select STYLES. You will see a list of all the styles available on your screen. To view the template code, double click on the name of the style. Select and copy the code you would like to edit from this window into the editor window. Below is the part of the SASWEB template style code that I will change using the TEMPLATE procedure in the editor window.

```
STYLE color_list
  "Colors used in the default style" /
    'fgD1' = cx666666
    'fgC1' = cxCCCCCC
    'fgB1' = cx000000
    'bgA1' = cx6495ED
    'fgA'  = cx003399
    'bgA'  = cxffffff;
```

The 'bgA1' = cx6495ED is the background blue color, and the 'fgA' = cx003399 is the blue font color in the SASWEB style. I used the Web site <http://www.colorschemer.com/online.html> to determine what these cx colors look like on my computer. To create a new style from the SASWEB style, use the code below. Edit the cx values for a custom color combination. The code below will change the blue font and blue background to shades of purple. I selected these shades of purple from the Colorschemer Web site. This Web site is a convenient way to get HEX numbers. The TEMPLATE procedure allows you to name your new custom style with the DEFINE statement. The name of this new style is SASWEB_PURPLE. After running the TEMPLATE procedure below, the SASWEB_PURPLE style will be available to use on your computer.

```
PROC TEMPLATE;  DEFINE STYLE Styles.SASWEB_PURPLE;
  PARENT = Styles.SASWEB;
  REPLACE color_list /
    'fgD1' = cx666666
    'fgC1' = cxCCCCCC
    'fgB1' = cx000000
    'bgA1' = cxBD64ED
    'fgA'  = cx660099
    'bgA'  = cxffffff;
end;
```

```
run;
```

Figure 4 gives an example of this SASWEB_PURPLE STYLE.

FIGURE 4: SASWEB_PURPLE

This is the SASWEB_PURPLE STYLE

Obs	NAME	TITLE_DEPT
1	Wendy	Statistician
2	Amy	Statistician

To learn more about the TEMPLATE procedure, I recommend reading the SUGI30 paper by Lauren Haworth, titled [SAS® with Style: Creating your own ODS Style Template for PDF Output](#).

PROC REPORT

There are different ways to change the colors in the REPORT procedure. This paper only shows examples of two methods I used to create customized PDF reports with different background colors: The COMPUTE statement using the CALL DEFINE statement, and the style option in the DEFINE statement. Other than the BACKGROUND style attribute, which changes the background color, there is the FOREGROUND style attribute that will change the color of the text. Other style attributes that affect the layout like OUTPUTWIDTH can be edited using these methods.

THE CALL DEFINE STATEMENT

My favorite use of the compute statement is to change the colors of rows, columns, and individual cells on my table with the CALL DEFINE statement. I will demonstrate how to change the color of a row, then a column, and then a cell. The PROC REPORT examples are taken from a complex report I created to monitor quality measures for home health agencies. Below is an explanation of the code I used.

ROW

First, let us look at how to highlight the background color of a row. This example is based on a report I created where possibly two or three rows need to be highlighted in light green (HONEYDEW). The COLUMN statement lists the order the report variables need to be processed:

```
COLUMN Add_poa Outcome Outcome_type BaselineRA month,RiskAdj ip_goal
      ("HHA rank" rank_May07_tie)
      ("STAR" star_target1_report star_target2_report start2_date);
```

Use a separate COMPUTE statement for each report-item variable that you want the color customized. Conditional variables that are used to change the color (highlight) of the desired rows must be listed before the COMPUTE report-item variable in the COLUMN statement. Notice that the first two variables are necessary to indicate which rows will be highlighted, and neither variable is displayed on the table. The third variable listed is the one labeled Outcome Desc (Type) on the table. In the CALL DEFINE statement below, _ROW_ indicates that the entire row should have the style changed.

```
COMPUTE Outcome_type;
  IF Add_poa = 'Yes'
    or Outcome in("Acute Care Hospitalization"
                  "Improvement in Management of Oral Medications")
  THEN CALL DEFINE(_ROW_, "style", "STYLE=[BACKGROUD=HONEYDEW]");
ENDCOMP;
```

COLUMN

Now let us look at how to highlight the background color of the columns. This example is based on a report where the background colors of the columns are changed for different sections of the table. The variable ip_goal is a light pink (LAVENDERBLUSH); the variable for the rank is white (SNOW), which needs to be specified since I do not want any of the rows to be green in this column; and the STAR section is light blue (ALICEBLUE). In the CALL DEFINE statement, _COL_ indicates the entire column should have the style changed:

```

COMPUTE ip_goal;
  CALL DEFINE(_COL_, "style", "STYLE=[BACKGROUND=LAVENDERBLUSH]");
ENDCOMP;
COMPUTE rank_May07_tie;
  CALL DEFINE(_COL_, "style", "STYLE =[BACKGROUND=SNOW]");
ENDCOMP;
COMPUTE star_target1_report;
  CALL DEFINE(_COL_, "style", "STYLE =[BACKGROUND=ALICEBLUE]");
ENDCOMP;
COMPUTE star_target2_report;
  CALL DEFINE(_COL_, "style", "STYLE =[BACKGROUND=ALICEBLUE]");
ENDCOMP;
COMPUTE start2_date;
  CALL DEFINE(_COL_, "style", "STYLE =[BACKGROUND=ALICEBLUE]");
ENDCOMP;

```

Figure 5 has two rows highlighted in light green and the last five columns highlighted in light pink, white, and light blue using the compute blocks shown above.

FIGURE 5: CHANGE COLOR OF ROWS AND COLUMNS

Good Health HHA (047xxx)
Risk Adjusted Rates (%)

		Period Ending													STAR		
		JUL06	AUG06	SEP06	OCT06	NOV06	DEC06	JAN07	FEB07	MAR07	APR07	MAY07	Goal	HHA rank	1st Target	2nd Target	Date Set 2nd
Outcome Desc (Type)	Jul04 Baseline	%	%	%	%	%	%	%	%	%	%	%		in AR May07			
Acute Care Hospitalization (Risk Adjusted)	33.42	35.98	35.79	35.44	35.76	35.40	34.47	34.97	35.77	34.97	33.95	35.08	28.21	0	30.00	28.60	01/24/07
Any Emergent Care (Risk Adjusted)	23.09	22.24	22.08	21.12	21.72	21.63	21.86	22.25	23.23	22.85	22.48	23.59	.	0			
Discharge to Community (Risk Adjusted)	62.54	58.57	58.40	58.70	58.49	58.94	60.21	59.75	58.91	59.82	60.47	59.75	72.15	0			
Improvement in Ambulation/Locomotion (Risk Adjusted)	34.94	36.96	37.53	37.73	37.27	38.00	38.93	38.64	37.27	37.83	38.38	40.08	45.95	0			
Improvement in Bathing (Risk Adjusted)	57.29	56.25	56.12	56.39	56.98	56.56	56.61	55.79	55.31	56.19	55.29	57.93	68.41	0			
Improvement in Dyspnea (Risk Adjusted)	44.01	43.83	43.67	43.84	44.71	45.09	46.08	45.43	45.02	47.40	49.34	53.10	62.86	0			
Improvement in Management of Oral Medications (Risk Adjusted)	32.58	45.68	45.80	45.75	46.11	48.05	47.85	47.59	48.39	47.32	42.58	46.75	42.92	0	37.00	40.00	01/24/07
Improvement in Pain Interfering with Activity (Descriptive)	55.66	57.95	59.51	60.16	60.60	60.98	59.44	60.31	61.46	62.33	63.02	62.22	69.74	0			
Improvement in Status of Surgical Wounds (Descriptive)	78.94	82.05	81.86	82.31	81.96	82.29	82.47	82.72	80.60	84.16	82.61	83.39	86.94	0			
Improvement in Transferring (Risk Adjusted)	46.12	51.80	49.61	49.11	48.42	48.23	48.04	47.98	48.04	50.21	50.05	51.95	59.73	0			
Improvement in Urinary Incontinence (Risk Adjusted)	30.80	33.88	33.63	34.05	35.28	35.06	35.45	36.12	37.22	38.48	36.88	39.14	50.93	0			

CELL

What if you decide not to highlight the last three columns in light blue, but want to highlight cells in these three columns depending on their values? Use the CALL DEFINE statements shown above, where the _COL_ indicates the entire column should have the style changed, but add a conditional statement in the COMPUTE block so that the style of a cell is changed only if that cell meets the requirement specified. To change the BACKGROUND style of a non-blank cell to light blue, use this code below.

```

COMPUTE star_target1_report;
  IF star_target1_report ^= ''
  THEN CALL DEFINE(_COL_, "style", "STYLE=[BACKGROUND=ALICEBLUE]");
ENDCOMP;
COMPUTE star_target2_report;
  IF star_target2_report ^= ''
  THEN CALL DEFINE(_COL_, "style", "STYLE=[BACKGROUND=ALICEBLUE]");
ENDCOMP;

```

```

COMPUTE start2_date;
  IF start2_date ^= ''
  THEN CALL DEFINE(_COL_, "style", "STYLE=[BACKGROUND=ALICEBLUE]");
ENDCOMP;

```

In Figure 6, instead of highlighting a few rows all in one color, each cell with a monthly rate is highlighted depending on a performance variable (perform). I call Figure 6 my STOPLIGHT report, since a cell is colored red if the rate is worse than baseline, yellow if the rate is better than baseline but not met the goal yet, and green if the rate is better than the goal.

Notice in the COLUMN statement that the ACROSS variable (month) is followed by a comma and the variable(s) with values that depend on each unit of the ACROSS variable are listed after the comma. As in this example, when there are multiple variables, list all of them in parentheses after the comma. The months are displayed left to right across the page, the rate (%) is the value of the RiskAdj variable displayed in the cells, and the background color of these cells depend on the value of the perform variable. The COLUMN statement used for this STOPLIGHT example is shown here.

```

COLUMN Add_poa Outcome Outcome_type BaselineRA month,(perform RiskAdj)
ip_goal
  ("HHA rank" rank_May07_tie)
  ("STAR" star_target1_report star_target2_report start2_date);

```

Notice that the conditional variable needed in the COMPUTE statement (perform) is listed before the COMPUTE report-item variable (RiskAdj). The order in the COLUMN statement is important, since it is the order the REPORT procedure processes the data. This example is more complex since the variable RiskAdj is under the ACROSS variable (month). Therefore, the CALL DEFINE statement will not have the generic _COL_, which indicates that the style is to be applied to the RiskAdj variable column. There is more than one column with rates. There is a set of columns for each of the two variables listed under the ACROSS variable (perform and RiskAdj) for each value of the ACROSS variable (month).

The REPORT procedure only processes the data cell by cell in the order specified; therefore, the basic IF-THEN logic using the name of the variable (perform) to condition on cannot be used in the COMPUTE block. A unique IF-THEN conditional statement will be needed for each column displayed under the ACROSS variable (month). There also needs to be a conditional statement for each possible background color. Use _Cxx_ to identify the column, where xx is the column number. Use the COLUMN statement to determine column numbers. Count every variable in the COLUMN statement, except the ACROSS variable, whether or not it is displayed in the report. In this example, the first set of conditional statements reference columns 5 (for the perform value) and column 6 (for the RiskAdj value) under the first month value (JUL06), and the second set of conditional statements reference columns 7 and 8 for the second month value (AUG06). The pattern is continued for all columns created under the ACROSS variable. The last set of conditional statements in this compute block reference columns 25 and 26 for the column labeled MAY07. The colors used in this Stoplight report are red (lightRED), yellow (cxFFFF99), green (lightGREEN) or white (SNOW) depending on the value of the perform variable.

```

COMPUTE RiskAdj;
  IF _c5_="WORSE THAN BASELINE" THEN CALL DEFINE("_c6_", "style",
"STYLE=[BACKGROUND=lightRED]");
  IF _c5_="GOAL NOT MET" THEN CALL DEFINE("_c6_", "style",
"STYLE=[BACKGROUND=cxFFFF99]");
  IF _c5_="GOAL MET" THEN CALL DEFINE("_c6_", "style",
"STYLE=[BACKGROUND=lightGREEN]");
  IF _c5_ in("NO DATA" "") THEN CALL DEFINE("_c6_", "style",
"STYLE=[BACKGROUND=SNOW]");

  IF _c7_="WORSE THAN BASELINE" THEN CALL DEFINE("_c8_", "style",
"STYLE=[BACKGROUND=lightRED]");
  IF _c7_="GOAL NOT MET" THEN CALL DEFINE("_c8_", "style",
"STYLE=[BACKGROUND=cxFFFF99]");
  IF _c7_="GOAL MET" THEN CALL DEFINE("_c8_", "style",
"STYLE=[BACKGROUND=lightGREEN]");
  IF _c7_ in("NO DATA" "") THEN CALL DEFINE("_c8_", "style",
"STYLE=[BACKGROUND=SNOW]");

  ...

  IF _c25_="WORSE THAN BASELINE" THEN CALL DEFINE("_c26_", "style",
"STYLE=[BACKGROUND=lightRED]");

```

```

IF _c25_="GOAL NOT MET" THEN CALL DEFINE("_c26_", "style",
"STYLE=[BACKGROUND=cxFFFF99]");
IF _c25_="GOAL MET" THEN CALL DEFINE("_c26_", "style",
"STYLE=[BACKGROUND=lightGREEN]");
IF _c25_ in("NO DATA" "") THEN CALL DEFINE("_c26_", "style",
"STYLE=[BACKGROUND=SNOW]");
ENDCOMP;

```

Figure 6 has the cells in the columns labeled with a month colored with the STOPLIGHT colors described above.

FIGURE 6: CHANGE COLOR OF CELLS

Good Health HHA (047xxx)
Risk Adjusted Rates (%)

Outcome Desc (Type)	Jul04 Baseline	Period Ending											Goal	HHA rank in AR May07	STAR		
		JUL06	AUG06	SEP06	OCT06	NOV06	DEC06	JAN07	FEB07	MAR07	APR07	MAY07			1st Target	2nd Target	Date Set 2nd
Acute Care Hospitalization (Risk Adjusted)	33.42	35.98	35.79	35.44	35.76	35.40	34.47	34.97	35.77	34.97	33.95	35.08	28.21	0	30.00	28.50	01/24/07
Any Emergent Care (Risk Adjusted)	23.09	22.24	22.08	21.12	21.72	21.63	21.86	22.25	23.23	22.85	22.48	23.59		0			
Discharge to Community (Risk Adjusted)	62.54	58.57	58.40	58.70	58.49	58.94	60.21	59.75	58.91	59.62	60.47	59.75	72.15	0			
Improvement in Ambulation/Locomotion (Risk Adjusted)	34.84	36.96	37.53	37.73	37.27	38.00	38.93	38.64	37.27	37.83	38.38	40.08	45.95	0			
Improvement in Bathing (Risk Adjusted)	57.29	56.25	56.12	56.39	55.98	56.55	56.81	55.79	56.31	56.19	55.29	57.93	68.41	0			
Improvement in Dyspnea (Risk Adjusted)	44.01	43.83	43.67	43.64	44.71	45.09	46.08	45.43	45.02	47.40	49.34	53.10	62.86	0			
Improvement in Management of Oral Medications (Risk Adjusted)	32.58	45.88	45.80	45.75	46.11	48.05	47.86	47.59	48.39	47.32	42.58	46.75	42.92	0	37.00	40.00	01/24/07
Improvement in Pain Interfering with Activity (Descriptive)	55.66	57.95	59.51	60.16	60.00	60.98	59.44	60.31	61.46	62.33	63.02	62.22	69.74	0			
Improvement in Status of Surgical Wounds (Descriptive)	78.84	82.05	81.86	82.31	81.96	82.29	82.47	82.72	80.60	84.16	82.61	83.39	86.94	0			
Improvement in Transferring (Risk Adjusted)	46.12	51.80	49.61	49.11	49.42	49.23	49.04	47.99	48.04	50.21	50.05	51.95	59.73	0			
Improvement in Urinary Incontinence (Risk Adjusted)	30.80	33.88	33.03	34.05	35.28	35.06	35.45	36.12	37.22	38.48	38.88	39.14	50.93	0			

THE DEFINE STATEMENT

Another way to change the background color of an individual cell is to use the STYLE option in the DEFINE statement of PROC REPORT. To do this you will need to create a user-defined format. Here are two example formats. The first one, called color_fmt, is for a variable that has a Yes/No response coded as text (either a Y or N). The second format defined below is for a numeric variable that is a ratio.

```

PROC FORMAT;
VALUE $ color_fmt
    "Y" = "SNOW"
    "N" = "RED";

VALUE c_ratio
    LOW - 0 = 'RED'
    0 - 1 = 'YELLOW'
    1 - HIGH = 'GREEN'
    OTHER = 'SNOW';

RUN;

```

Unlike the previous example, where the color applied to a variable is dependent on another variable, when using the DEFINE statement, the color depends on the value of the DEFINE report-item variable. Only the cells in the column of the DEFINE variable that has the STYLE option will be effected by the attribute set to a user-defined format.

```
DEFINE flagifnotsame /GROUP MISSING CENTER "Same?"  
STYLE={BACKGROUND=$color_fmt.};  
  
DEFINE ratio /GROUP MISSING "Ratio"  
STYLE={BACKGROUND=c_ratio.};
```

Use the STYLE options in the DEFINE statement of the REPORT procedure to highlight cells that are of interest.

CONCLUSION

If your resources allow for the use of color in either printed or electronic format, I challenge you to find the STYLE(s) that works best for you. To highlight information, I prefer to customize the background color. Keep in mind that not only the BACKGROUND can be edited using these methods, but other STYLE attributes like FOREGROUND can also be customized. Use your color printer to show everyone what your STYLE looks like!

REFERENCES

Carter, Jeff . 2005. "Use of Styles in Graphics." Proceedings of the thirtieth annual SAS® Users Group International Conference, Philadelphia PA.
<http://support.sas.com/rnd/datavisualization/papers/sugi30/GraphStyles.pdf>

Code extract from SAS 9.1.3 Service Pack 4 Copyright (c) 2002-2003 by SAS Institute Inc., Cary, NC, USA.

Colorschemer Online v.2. <http://www.colorschemer.com/online.html>. Accessed 12/28/2006.

Haworth, Lauren. 2005. "SAS® with Style: Creating your own ODS Style Template for PDF Output." Proceedings of the thirtieth annual SAS Users Group International Conference, Philadelphia PA, 132-30.
<http://www2.sas.com/proceedings/sugi30/132-30.pdf>

The data displayed in the examples are fictitious and should only be used for report layout examples.

ACKNOWLEDGMENTS

Special thanks to everyone at AFMC for all their helpful advice and support.

The content of this publication does not necessarily reflect the views or policies of Arkansas Foundation for Medical Care (AFMC). The author assumes full responsibility for the accuracy and completeness of the ideas presented.

RECOMMENDED READING

For more information about quality measures in the home health setting, go to the [Home Health Compare](http://www.medicare.gov/HHCompare/Home.asp?dest=NAV|Home|About#TabTop) Web site. <http://www.medicare.gov/HHCompare/Home.asp?dest=NAV|Home|About#TabTop>

CONTACT INFORMATION

Your comments and questions are valued and encouraged. Contact the author at:

Wendy Boberg
Statistician, Data Mining Team Leader
Arkansas Foundation for Medical Care (AFMC)
401 W Capitol Ave., Suite 508
Little Rock, AR 72201
Phone: 501-212-8716
Fax: 501-375-5705
E-mail: wboberg@afmc.org
Web: <http://www.afmc.org>

SAS and all other SAS Institute Inc. product or service names are registered trademarks or trademarks of SAS Institute Inc. in the USA and other countries. ® indicates USA registration.

Other brand and product names are trademarks of their respective companies.